CS 110, Section 3 – Exam 1 Review

- ***What is printed when each program is executed?***
  - *Note, not all programs compile. Some have errors.*
- ***Explain why the output is generated, or what caused the error.***

1.
```
int i = 10;
println(i);
```

2.
```
int j;
println(j);
```

3.
```
int j;
j = 0;
println(j);
```

4.
```
float x = -10;
println(x);
```

5.
```
double dx = 9876543210.0;
println(dx);
```

6.
```
boolean b = false;
println(!b);
```

7.
```
boolean b2;
b2 = true;
println(b2);
```

8.
```
String name = "Tiki";
println( name );
```

9.
```
int i = 10;
int j = 20;
float k = i/j;
println( k );
```

10.
```
float x = 10.0;
float y = 20.0;
int z = x/y;
println(z);
```

11.
```
float x = 30;
int y = 10;
float z = x/y;
println( z );
```

12.
```
int x = 40;
float y = 30;
float z = x/y;
println( z );
```

13.
```
int i = 60;
int j = 40;
int k = i/j;
println( k );
```

14.
```
int i = 2;
i++;
println(i);
```

15.
```
int i = -2;
i++;
println( i );
```

16.
```
int j = 3;
j--;
println(j);
```

17.
```
int j = -2;
j--;
println( j );
```

18.
```
float x = 12.0;
x += 2;
println(x);
```

19.
```
float y = 7.0;
y += y;
println( y );
```

20.
```
float x = 6;
float y = 2;
x -= y;
println( x );
println( y );
```

21.
```
int m = 2;
int n = 2;
int o = 2;
m++;
n += 1;
o = o + 1;
println(m);
println(n);
println(o);
```

22.
```
float x = 12.0;
x /= 3.0;
println(x);
```
23.
```
float y = 5.0;
y *= 3.0;
println(y);
```

24.
```
int j = 9;
println( j % 9 );
println( j % 8 );
println( j % 7 );
println( j % 6 );
println( j % 5 );
println( j % 4 );
println( j % 3 );
println( j % 2 );
println( j % 1 );
println( j % 0 );
```

25.
```
int k = 12.0;
println( k );
```

26.
```
int l = int(12.0);
println( l );
```

27.
```
int j = 4 + int("3");
println(j);
```

28.
```
println( 3 + "4" );
```

29.
```
boolean b = (10 < 5);
println( b );
```

30.
```
boolean b = (-10 > -5);
println( b );
```

```
31. boolean b = (10 <= 10);
    println( b );

32. boolean b = (-1.1 >= -1.2);
    println( b );

33. int b = (3 != 3);
    println( b );

34. boolean b = (4.0 == 4);
    println( b );

35. boolean b = ("ABC" == "ABC");
    println( b );

36. boolean b = ("ABC" != "abc");
    println( b );

37. boolean b = ("B" > "A");
    println( b );

38. int b = "B".compareTo("A");
    println( b );

39. boolean b = "B".compareTo("b");
    println( b );

40. String s1 = "A";
    String s2 = "B";
    println( s1.compareTo( s2 ) );

41. println( !true );

42. println( !!true );

43. boolean b = (true && true);
    println( b );

44. boolean b = (true || true);
    println( b );

45. boolean b = (!true || true);
    println( b );

46. boolean b = (true || !false);
    println( b );
```

47.
```
int d = 6;
boolean b = (d!=0 && d!=6);
println( b );
```

48.
```
int d = 6;
boolean b = (d!=0 && d!=6);
println( b );
```

49.
```
int d = 6;
boolean b = !(d==0 || d==6);
println( b );
```

50.
```
int d = 6;
boolean b = !(d==0 && d==6);
println( b );
```

51.
```
int y = 0;
boolean b = (y < 0 || y > 0);
println( b );
```

52.
```
int y = 1;
boolean b = (y < 0 || y > 0);
println( b );
```

53.
```
int y = 1;
boolean b = (y < 0 && y > 0);
println( b );
```

54.
```
int k = 11;
if ( k > 10 ) {
  println( "true" );
}
```

55.
```
int max = 100;
int min = 10;
int temp = 50;
if ( temp > max ) {
  println("too hot");
} else if ( temp < min ) {
  println("too cold");
} else {
  println("just right");
}
```

56.
```
int max = 100;
int min = 10;
int temp = 5;
if ( temp > max && temp < min ) {
  println("too hot or too cold");
} else if ( temp >= min || temp <= max ) {
  println("just right");
}
```

57.
```
int max = 100;
int min = 10;
int temp = 5;
if ( temp > max )  println("too hot");
if ( temp < min )  println("too cold");
if ( temp >= min ) println("ok 1");
if ( temp <= max ) println("ok 2");
```

58.
```
int x1 = 20;
int y1 = 30;
int x2 = 100;
int y2 = 110;
int mouseX = 50;
int mouseY = 50;
if (mouseX < x1)      println("not in the rect");
else if (mouseY < y1) println("not in the rect");
else if (mouseX > x2) println("not in the rect");
else if (mouseY > y2) println("not in the rect");
else                  println("must be in the rect");
```

59.
```
String name = "Spongebob";
if (name == "Patrick") {
  println("I am a starfish");
} else if (name == "Gary") {
  println("I am a snail");
} else if (name == "Squidward") {
  println("I am a squid");
} else if (name == "Spongebob") {
  println("I am a sponge");
} else {
  println("I do not live in Bikini Bottom");
}
```

60.
```
char k = 'B';
switch(k) {
  case 'A':
    println("A");
    break;
  case 'B':
    println("B");
  case 'C':
    println("C");
    break;
  default:
    println("something else");
}
```

61.
```
for (int i=0; i<3; i++) {
  println(i);
}
```

62.
```
for (int i=0; i<5; i+=2) {
  println(i);
}
```

63.
```
for (int i=0; i<5; i=i+1) {
  if ( i % 2 == 0) {
    println( i + " is even" );
  } else {
    println( i + " is odd" );
  }
}
```

64.
```
for (int j=3; j>0; j--) {
  println(j);
}
println("lift off");
```

65.
```
for (int i=0; i<100; i++) {
  if (i < 51) continue;
  if (i > 54) break;
  println(i);
}
```

66.
```
int k = 0;
while (k < 3) {
  println(k);
  k++;
}
```

67.
```
int k = 3;
while (k >= 0) {
  println(k);
  k--;
}
```

68.
```
int k = 0;
while (k < 100) {
  k++;
  if (k < 51) continue;
  if (k > 54) break;
  println(k);
}
```

69.
```
int i=0;
while (i<5) {
  if ( i % 2 == 0) println( i + " is even" );
  else             println( i + " is odd" );
  i=i+1;
}
```

70.
```
while (true) {
  float x = random(100);
  if (x < 10.0) break;
  println( x + " is >= 10");
}
```

71.
```
for (int i=0; i<2; i++) {
  for (int j=0; j<3; j++) {
    println(i + "," + j);
  }
}
```

72.
```
int i=0;
while (i < 2) {
  int j = 0;
  while (j < 3) {
    println(i + "," + j);
    j++;
  }
  i++;
}
```

73.
```
int i=0;
while (i < 2) {
  for (int j=0; j<3; j++) {
    println(i + "," + j);
  }
  i++;
}
```

74.
```
int[] vals;
vals = new int[3];
for (int i=0; i<vals.length; i++) {
  println( vals[i] );
}
```

75.
```
int[] things = new int[4];
for (int i=0; i<things.length; i++) {
  println( things[i] );
}
```

76.
```
float[] stuff = new float[4];
for (int i=0; i<stuff.length; i++) {
  stuff[i] = random(10);
}
for (int i=0; i<stuff.length; i++) {
  println( stuff[i] );
}
```

77.
```
int[] tri = new int[5];
for (int i=0; i<tri.length; i++) {
  tri[i] = int(random(10));
}
for (int j=tri.length-1; j>=0; j--){
  for (int i=0; i<=j; i++) {
    print( tri[i] );
    print( " " );
  }
  println();
}
```

78.
```
void setup() {
  printInt(9);
}
void printInt(int i) {
  println(i);
}
```

79.
```
void setup() {
  printInt(9.5);
}
void printInt(int i) {
  println(i);
}
```

80.
```
void setup() {
  float x = 3.1;
  x = squareIt(x);
  println(x);
}
float squareIt(float x) {
  x *= x;
  return x;
}
```

81.
```
void setup() {
  println( squareIt(3.1) );
}
float squareIt(float x) {
  return x*x;
}
```

82.
```
void setup() {
  checkMagic(2.1);
  checkMagic(PI);
}
void checkMagic(float k) {
  if ( isMagic(k) == true )
    println(k + " is magic");
  else
    println(k + " is not magic");
}
boolean isMagic(float m) {
  if (m == PI) {
    return true;
  } else {
    return false;
  }
}
```

83.
```
void setup() {
  float x = 64.0;
  float y = 24.0;
  float g = GCD(x, y);
  println("The GCD of " + x + " and " + y + " is " + g);
}
float GCD(float A, float B) {
  while (B != 0) {
    if (A > B) {
      A = A - B;
    } else {
      B = B - A;
    }
  }
  return A;
}
```

84.
```
void setup() {
  float[] inseams = createRandomArray(2);
  float[] hatSizes = createRandomArray(3);
  printArray( inseams );
  printArray( hatSizes );
}
float[] createRandomArray(int size) {
  float[] arr = new float[size];
  for (int i=0; i<arr.length; i++) {
    arr[i] = random(1.0);
  }
  return arr;
}
void printArray(float[] arr) {
  for (int i=0; i<arr.length; i++) {
    println( arr[i] );
  }
}
```

85.
```
float x = 1.2;
void setup() {
  println(x);
}
```

86.
```
float x = 1.2;
void setup() {
  float x=3.4;
  println(x);
}
```

87.
```
float x = 1.2;
void setup() {
  float x=3.4;
  printIt();
  println(x);
}
void printIt() {
  println(x);
}
```

88.
```
float x = 1.2;
void setup() {
  x=3.4;
  printIt();
  println(x);
}
void printIt() {
  println(x);
}
```

89.
```
float x = 1.2;
void setup() {
  float x=3.4;
  printIt(x);
  println(x);
}
void printIt(float y) {
  println(x);
}
```

90.
```
String s1 = "blah";
String s2 = new String("blah");
int b = s1.compareTo(s2);
println(b);
```

91.
```
void setup() {
  Emotion e1;
  e1.print();
}

class Emotion {
  String name = null;
  int strength = 5;

  Emotion(String _name, int _strength) {
    name = _name;
    strength = _strength;
  }

  void print() {
    if (strength > 5) println("I feel a lot of " + name);
    else              println("I feel a little " + name);
  }
}
```

92.
```
void setup() {
  Emotion e1;
  e1 = new Emotion("happiness", 9);
  e1.print();
}

class Emotion {
  String name = null;
  int strength = 5;

  Emotion(String _name, int _strength) {
    name = _name;
    strength = _strength;
  }

  void print() {
    if (strength > 5) println("I feel a lot of " + name);
    else              println("I feel a little " + name);
  }
}
```

93.
```
void setup() {
  Emotion e1 = new Emotion("happiness", 9);
  Emotion e2 = new Emotion("anger", 2);
  e1.print();
  e2.print();
  e2.strength = e1.strength;
  e2.print();
}

class Emotion {
  String name = null;
  int strength = 5;

  Emotion(String _name, int _strength) {
    name = _name;
    strength = _strength;
  }

  void print() {
    if (strength > 5) println("I feel a lot of " + name);
    else              println("I feel a little " + name);
  }
}
```

94.
```
void setup() {
  Emotion e1 = new Emotion("happiness", 9);
  e1.print();
  e1.name = "asparagus";
  e1.print();
}

class Emotion {
  String name = null;
  int strength = 5;

  Emotion(String _name, int _strength) {
    name = _name;
    strength = _strength;
  }

  void print() {
    if (strength > 5) println("I feel a lot of " + name);
    else              println("I feel a little " + name);
  }
}
```

95.
```
void setup() {
  int num = int(random(30));
  Child[] children = new Child[num];
  println("I have " + children.length + " children");
}
class Child {
  int age;
  Child() {
    age = int(random(20));
  }
}
```

96.
```
void setup() {
  int num = int(random(10));
  Child[] children = new Child[num];
  println("I have " + children.length + " children");
  println("Their ages are:");
  for (int i=0; i<num; i++) println(children[i].age);
}
class Child {
  int age;
  Child() {
    age = int(random(20));
  }
}
```

97.
```
void setup() {
  int num = int(random(10));
  Child[] children = new Child[num];
  for (int i=0; i<num; i++) children[i] = new Child();
  println("I have " + children.length + " children");
  println("Their ages are:");
  for (int i=0; i<num; i++) println(children[i].age);
}
class Child {
  int age;
  Child() {
    age = int(random(20));
  }
}
```

98.
```
void setup() {
  Parent p1 = new Parent();
  p1.print();
}
class Parent {
  Child[] children;
  Parent() {
    int num = int(random(10));
    children = new Child[num];
    for (int i=0; i<num; i++) children[i] = new Child();
  }
  void print() {
    println("I have " + children.length + " children");
    println("Their ages are:");
    for (int i=0; i<children.length; i++) println(children[i].age);
  }
}
class Child {
  int age;
  Child() {
    age = int(random(20));
  }
}
```

99.
```
float x = 1.2;
void setup() {
  println(x);
  float x = 3.4;
  println(x);
  Tree t1 = new Tree(x);
  println( t1.x );
  printIt();
}
class Tree {
  float x=5.6;
  Tree(float tx) {
    println(x);
    x = tx;
  }
}
```

100.
```
float x = 1.2;
 void setup() {
  println(x);
  float x = 3.4;
  println(x);
  Tree t1 = new Tree(x);
  println( t1.x );
  printIt();
  printIt();
}
class Tree {
  float x=5.6;
  Tree(float tx) {
    println(x);
    x = tx;
  }
}
void printIt() {
  println(x);
  x=7.8;
}
```